
News  

Spike2 and Signal Training Days 
Our U.K. Spike2 and Signal training courses will take place on Thursday 30th April and 
Friday 1st of May at St. John’s Innovation Centre, Cambridge. These training days are for 
both existing and prospective users of Spike2 and Signal and are suitable for beginners 
and advanced users. There is also our popular Workshop thread where you can discuss 
your particular requirements with a CED engineer or programmer in a small group 
environment. 

If you or any of your colleagues would like to attend, please follow the links to view the 
training day program and complete the registration form. Please give a short description 
of the type of research that you do and analysis that you are interested in. This will allow 
us to tailor sessions for specific requirements. 

Training Days in Shanghai 
We will also be running training days in Shanghai during the week beginning the 6th of 
April. These training days will be run in association with Qichi Instruments, distributors of 
CED equipment in China. For further information please contact Sales@ced.co.uk 

The CED Forum 
The CED forum has now been running for 6 months and has over 100 members. 
Registered users can post questions about software and hardware for the CED team, 
suggestions for future products and software revisions and browse generally useful 
information posted by ourselves and other CED users. 

Meetings and events 
Australian Neuroscience Society 
National Convention Centre 
Canberra, Australia 
January 27th – 30th 2009 

Biophysical Society Annual Meeting 
Boston Convention and Exhibition Center  
Boston, MA, USA 
March 1st – 3rd 2009 

German Neuroscience Society  
Goettingen Exhibition Centre   
Goettingen, Germany 
March 26th – 29th 2009 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.10 Signal version 4.03 

Spike2 version 5.20 Signal version 3.10 

Spike2 demo Signal demo 

 

Spike2   
 

Q. I use active cursors for off-line measurements and was wondering if there was 
any way to copy settings between data files, rather than having to define the 
cursor settings for every file I analyse? 

A. Active cursor and display settings for a data file are stored in an associated 
resource file, with the file extension .S2R, which is saved in the same location as 
the data file and with the same name so that the file can be re-opened with 
exactly the same screen appearance and cursor settings it had when the file was 
closed. You can apply any resource file to a data file in Spike2 from the File 
menu Resource Files option. The resource file including the cursor settings that 
you want to use can be applied manually or can be set as a Global resource that 
is applied automatically. 

This example will apply settings from a single resource file to all data files in a 
given folder. Open the resources dialog and check the ‘Use global resource files’ 
checkbox. Type the name of the resource file you want to use (this will have the 
same name as the file you want to copy the settings from) and set the resource 
file location from the drop-down list. Additional fields in the dialog allow you to 
specify which files should have the resources applied. Set the data file path to 
only apply settings to files in a single folder. 

 
Global resources dialog 

  

 

Q. When sampling with Signal, the data file always displays the last recorded frame 
when sampling is stopped. I would much prefer that the file displayed the first 
saved frame by default.  

A. The Edit menu Preferences dialog in Signal is a tabbed dialog that includes a 
range of different options for the display, sampling and export of data. The 
Compatibility tab of the dialog contains a checkbox to set the option of displaying 
Frame 1 rather than Frame 0 once sampling is stopped. 

When using Active Cursors on-line, the settings are saved in the sampling configuration once sampling is stopped. Subsequent data files sampled with this configuration will use the previous cursor settings



 
Edit preferences dialog 

Further information on preference settings can be found in the Signal on-line help. 

Scripts: Spike2  
Q. Is there a quick and easy way to copy the display of visible/hidden channels from 

one data file to another in Spike2? 

A. The attached pair of scripts, GetVisCh.s2s and SetVisCh.s2s, provide a 
quick way of applying the Show / Hide state of channels in one view to others 
and should be saved in your Spike6\Scripts folder before use. Both scripts add a 
button to the Spike2 Script bar and display it when first run for “one click” 
operation. Simply click on the GetVisCh button in the script bar to copy the layout 
of the front view, then click on the destination view to bring it to the front and click 
the SetVisCh button to apply the stored Show /Hide channel settings. You can 
optionally hold down Ctrl when pressing SetVisCh to apply, as far as possible, 
all the characteristics of the source (time ranges, trace colours etc.) to the 
destination. This script is for Spike2 version 6 only and makes use of the 
GHUtils.s2s script that is stored in the Spike6\include folder during installation. 

Scripts: Signal  
Q. I would like an option for text output of Signal data that would print the data from 

each frame in a separate column for a selected channel. This would make it 
much easier to perform cross-frame analysis in a spreadsheet package. 

A. The attached script, FrameCols.sgs, is an example of exporting a single 
channel from a Signal file so that the data from each frame is printed in a 
separate column. The user can select which channel to export from a dialog 
displayed when the script is run. 

Did you know…? 
The CED 1401 device drivers are now digitally signed. This allows Signal and Spike2 to 
be used on a PC running Microsoft Vista x64 and reduces the number of queries when 
installing in 32-bit Vista. 

Recent questions   
Q. I am analysing large files that take a long time to display when I open them in 

Spike2. Is there a way to stop them drawing all of the data in the file? 

A. You can break out of long drawing or calculation operations in Spike2 using the 
Ctrl+Break keyboard shortcut. This is useful if you have inadvertently 
displayed or analysed a very large section of data. There is also an option in the 
General tab of the preferences dialog to ignore resource file X range settings to 
give a fast initial display by only showing the first second of data when opened. 

User group      
Try the new CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

The Clamp settings are available in the Preferences dialog if the Features for Patch and Voltage clamp experiments option was selected during the Signal installation.


'$GetVisCh|Store the layout of the front view. Click on SetVisCh to apply it to another data file.
'===========================================================================================================
'	          CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB6 0FE, UK
'===========================================================================================================

'Copyright © Cambridge Electronic Design. last modified:  11/Apr/2008.
'Author:	BGH.

'SOFTWARE REQUIRED:
'Spike 2 version 6.06 or higher
'For correct performance you must store this script in the <scripts> folder of the Spike2 directory
'This script uses functions in GHutils.s2s located in a folder called <include> in the Spike2 root directory

'HEALTH WARNING:
'The script is a <work in progress> and is offered without guarantees.
'Please report any problems to:	geoff@ced.co.uk

'OVERVIEW
'This script together with its partner SetVisible.s2s provides a quick way of copying the layout of one view to another.
'By default, the scripts get and set the list of visible channels in the destination file to be the same as the source file.
'You also have the option to apply the resources of the source to the destination file.
'This has the effect of matching all the characteristics of the file, channels, axis ranges, drawing modes, colours etc. as closely as possible.

'USER GUIDE
'This script is one of a pair with complementary roles:
'The <GetVisible.s2s> script saves a list of visible channels (and optionally many other characteristics) in the Front View.
'The <SetVisible.s2s> script applies this layout to the current Front View.
'When you run these scripts for the first time, they will add buttons to the Script Bar for easy access in the future.

'Simply click on the source data file to bring it to the front and then press the Script Bar button labelled <GetVisCh> to save the current layout.
'Click on a destination data file and press the button labelled <SetVisCh> in order to apply the stored list of channels to this view.
'Hold down the <Ctrl> key when pressing the <SetVisCh> button to copy all of the characteristics of the source file to the destination.
'as far as possible. Note that channels in the destination file that are incompatible with the corresponding channel in the source will be hidden.
'You can make them visible again by clicking <SetVisCh> again without holding down <Ctrl>.

'You can apply a format to multiple destination files by bringing each one to the front in turn and clicking
'on <SetVisCh> or Ctrl+<SetVisCh>.

#include <GHutils.s2s>	                               'include functions stored in  include\GHutils.s2s (inside the Spike2 folder)

var rootpath$;
var set$;

rootpath$:=FilePath$(2);	                            'path to Spike2 application
set$:="GetVisCh|"+rootpath$+"\scripts\\GetVisible.s2s|Get the Show/Hide state of channels in the Front View";	'label for script bar button
AddScriptBarBtn%(set$);
DoGetVisible();
halt

proc DoGetVisible();  'show script bar
var vh%,vkind%,txt%,i%;
var	rfname$,ok%;

vh%:=FrontView(); 
vkind%:=ViewKind(vh%);
if vkind% <> 0 and vkind% <> 4 and vkind% <> 12 then
	NewsFlash("Front view is not a Time, Result or XY view",0,0,0,1);
else
	View(vh%);
	var n%,lst%[2];
		for i%:= 1 to 4 do
			rfname$+=FileName$(i%);	                      'build up full file name and path for resource file
		next;
		rfname$+=".s2r";
		ok%:=FileSaveResource(rfname$);	                'make sure we have an up to date resource file for this view
	n%:=ChanList(lst%,2048);	                         'size array for visible channels
	var chlst%[n%+1];
	ChanList(chlst%,2048);	                            'fill array
	
	txt%:=FileOpen(Print$("%sscripts\\showhide.txt",rootpath$),8,1);	'open external text file for writing in Spike2 scripts folder
	if txt% < 0 then
		NewsFlash(Error$(txt%),3.0,0,0,1);
		return;
	endif;
	if ok%=0 then	                                     'if we saved a resource file 
		Print("%s\n",rfname$);	                         'print name of resource file to text file
	else
		Print("x");
	endif;
	for i%:=0 to n% do	
		Print("%d\n",chlst%[i%]);	                      'save channel list one channel per line starting with channel count
	next;
	FileClose();
FrontView(vh%);	                                     'restore selected view to front	
endif;
return;
end;

'$SetVisCh|Apply a layout stored using GetVisCh to the front view.
'===========================================================================================================
'	          CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB6 0FE, UK
'===========================================================================================================

'Copyright © Cambridge Electronic Design. last modified:  11/Apr/2008.
'Author:	BGH.

'SOFTWARE REQUIRED:
'Spike 2 version 6.06 or higher
'For correct performance you must store this script in the <scripts> folder of the Spike2 directory
'This script uses functions in GHutils.s2s located in a folder called <include> in the Spike2 root directory

'HEALTH WARNING:
'The script is a <work in progress> and is offered without guarantees.
'Please report any problems to:	geoff@ced.co.uk

'OVERVIEW
'This script together with its partner GetVisible.s2s provides a quick way of copying the layout of one view to another.
'By default, the scripts get and set the list of visible channels in the destination file to be the same as the source file.
'However, you also have the option to apply the resources of the source to the destination file.
'This has the effect of matching all the characteristics of the file, channels, axis ranges, drawing modes, colours etc. as closely as possible.

'USER GUIDE
'This script is one of a pair with complementary roles:
'The <GetVisible.s2s> script saves a list of visible channels (and optionally many other characteristics) in the Front View.
'The <SetVisible.s2s> script applies this layout to the current Front View.
'When you run these scripts for the first time, they will add buttons to the Script Bar for easy access in the future.

'Simply click on the source data file to bring it to the front and then press the Script Bar button labelled <GetVisCh> to save the current layout.
'Click on a destination data file and press the button labelled <SetVisCh> in order to apply the stored list of channels to this view.
'Hold down the <Ctrl> key when pressing the <SetVisCh> button to copy all of the characteristics of the source file to the destination,
'as far as possible. Note that channels in the destination file that are incompatible with the corresponding channel in the source will be hidden.
'You can make them visible again by clicking <SetVisCh> again without holding down <Ctrl>.

'You can apply a format to multiple destination files by bringing each one to the front in turn and clicking
'on <SetVisCh> or Ctrl+<SetVisCh>.

#include <GHutils.s2s>	                               'include functions stored in  include\GHutils.s2s (inside the Spike2 folder)

var rootpath$;
var set$;

rootpath$:=FilePath$(2);	                           	'path to Spike2 application
set$:="SetVisCh|"+rootpath$+"\scripts\\SetVisible.s2s|Get the Show/Hide state of channels in the Front View";	'label for script bar button
AddScriptBarBtn%(set$);
Seconds(0);
var chr%;
repeat
	chr%:=Inkey();
until chr%=17 or Seconds()>0.1;	                    	'check whether the Ctrl key was pressed
DoSetVisible(chr%);
halt

proc DoSetVisible(ctrl%);
var vh%,vkind%,txt%;
var xl,xh,yl,yh;
var rfname$,ok%;

vh%:=FrontView(); 
vkind%:=ViewKind(vh%);
if vkind% <> 0 and vkind% <> 4 and vkind% <> 12 then
	NewsFlash("Front view is not a Time, Result or XY view",0,0,0,1);
	return;
else
	var n%,i%;
	txt%:=FileOpen(Print$("%sscripts\\showhide.txt",rootpath$),8,0);	'open external text file for reading
	if txt% < 0 then
		NewsFlash("Textfile holding a list of channels to show was not found.",3.0,0,0,1);
		return;
	endif;
	Read(rfname$);	                                 	'read name of resource file -if any.
	Read(n%);	                                       	'read first value
	var chlst%[n%+1];	                               	'size array to hold results
	chlst%[0]:=n%;
	for i%:=1 to n% do
		Read(chlst%[i%]);
	next;
	FileClose();

	if ctrl%=17 and InStr(rfname$,".s2r") > 0 then	 	'ctrl pressed, time view and resource file available
		View(vh%);
		WindowGetPos(xl,yl,xh,yh);	                 	'get current window position
		ChanHide(-1); 'hide all initially
		ok%:=FileApplyResource(rfname$);	           	'apply resources from source file
		if ok%>0 then
			Window(xl,yl);	                         	'revert to the original window position
			FileSaveResource(1);	                   	'save updated resource file
		else
			View(vh%);
			ChanHide(-1);
			ChanShow(chlst%[]);	                        	'just update the channel list			
		endif;
	else
		View(vh%);
		ChanHide(-1);
		ChanShow(chlst%[]);	                        	'just update the channel list
	endif;	
endif;
FrontView(vh%);	                                    	'restore selected view to front
return;
end;

'$FrameCols.sgs|Example script to print text output of selected channel data with frames in
'separate columns

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 12/01/09

Var data%;                          ' View handle for our data file
var XCol%:=1;                       ' Select X (time) columns - set to one for times or zero for no times
Var start:=0;                  ' This is the start time within each frame
Var endpos:=1;                  ' This is the end time within each frame
Var chan% := 1;                     ' Tne channel number to export

Data% := Frontview();

DlgCreate("Select channel");
DlgChan(1,"Select channel",1);
DlgReal(2,"Start",0,100000);
DlgReal(3,"End",0,100000);
DlgShow(chan%,start,endpos);
DoExport();
halt;

proc DoExport();
var fp%;                            ' Index for first point wanted
Var x%;                             ' Count of points per frame - sets number of rows
Var y%;                             ' Count of colums to generate

Frame(1);
x% := Chanrange(chan%, Start, endpos, fp%); ' Find out how many points and frames
y% := FrameCount()+XCol%;
if ((x% * y%) > 20000000) then      ' A safety check for too much memory used,
    Message("Too many data points and frames"); ' you can disable this by commenting out
    return;                         ' but at your own risk!
endif

var f%, i%;
Var Export[x%][y%];                 ' The array to hold all the data

if (XCol% > 0) then                 ' Firstly, if we want a times column
    for i% := 0 to x%-1 do              '  build the list of times
        Export[i%][0] := BinToX(chan%, fp%+i%);
    next;
endif;

PrintLog("Time\t");
for f%:= 1 to Framecount() do       ' Next copy all of the data into the array
    Frame(f%);                      ' Select the next frame
    ArrConst(Export[][f%],View(data%,chan%).[fp%:x%]);  ' and copy the data
    PrintLog("Frame %d\t",f%);
next
PrintLog("\n");
for i% := 0 to x%-1 do              ' Now generate the text output - this loops through the rows
    for f% := 0 to y%-1 do          ' This loops through the columns
        if (f% > 0) then PrintLog("\t") endif;  ' Tab separator
        PrintLog("%8g", Export[i%][f%]);
    next;
    PrintLog("\n");                 ' New line after each row
next;
end

